

Personal Computing

\$2.00
JUNE 1978



**PARTNER
MATCHING
WITH YOUR MICRO**



page 21



page 39



page 50

Cover illustration
by Doug Smith
Cover photo
by Jon Buchbinder

Cover Story

MATCHING PARTNERS 54

Instead of leaving things to chance, be it Cupid's arrow or names pulled out of a hat, with this program you can match compatible tennis partners, equally talented basketball teams or just two people with the same interests. *by Harriet Morrill*

LAUNCHING PAD

MICROCOMPUTERS IN ONE EASY LESSON 21

Plunging into computer literature can leave the novice as befuddled as trying to read Shakespeare in Greek. This easy-to-understand computer primer explains the basics in simple terms — and includes hints on selecting a system as well.

DIGGING IN

CONTRACT FULFILLMENT 27

Keeping track of which customers fulfill their contracts requires hours of time and complex record-keeping. This program shows how to let your computer do the work for you. *by Sam Newhouse*

PICKING A PRINTER 60

Printers for micros offer a wide range of features, specifications and prices. Here we review these factors to make your selection easier. *by Chip A. Tyeti*

RELOCATABLE ROUTINES 76

Are you tired of writing and rewriting your favorite routines to fit each individual program? You can save needless programming time and effort by using relocatable BASIC routines. *by Robert Irving*

ON THE LIGHTER SIDE

PROGRAMMING THE MOST COMPLEX COMPUTER 26

Learn how to program the most complex computer in the universe — your brain.

FACTOR GAME 44

Sharpen your math skills while having fun with this easy-to-learn game. *by Herbert L. Dershem*

DOT BY DOT 69

You probably played this game back in ninth grade study hall and consistently beat your friends; now you can play against your computer — and lose! *by Fred Helliwell*

MAKING MONEY

PRINTING CONTRACT FORMS 39

With this program you can print contracts for your own business or sell custom-made contracts to local builders, small businesses and contractors. *by Charles A. Matz*

INSTANT REPLAY

INSIDE YOUR MICROCOMPUTER 79

What actually goes on inside a computer? Join us on a guided tour through your micro's innards to find out.

FEEDBACK	4
RANDOM ACCESS	9
ONCE UPON A TIME	50
COMPUTER CHESS	87
PRODUCT CLOSE-UP	99
WHAT'S COMING UP	101

Factor Game

BY HERBERT L. DERSHEM

With this educational game you can sharpen your skills in factoring integers and develop analytical problem solving ability. The game, Factor, is designed for anyone with seventh grade math level ability. But bright youngsters should be able to play the game without too much difficulty.

The game was first described by J.B. Harkin and D.S. Martin in the November 1973 issue of *The Arithmetic Teacher*. This version of Factor adds to the flexibility of the original and can be implemented on a microcomputer system.

Program Run 1 shows a sample game. The underlined portions are typed by the user, who plays against the computer.

The computer first asks for the name of its opponent and then lets him choose the size of the array of numbers. Any size from 2 to 81 will be accepted, although arrays smaller than 10 don't make for very exciting games.

The user has the option of choosing first or letting the computer choose first. Only the very naive or the very confident will let the computer choose first, since the first player has an advantage. However, the larger the array, the smaller the advantage.

The computer then prints the array, which consists of all integers from 2 to the limit entered by the user, and invites the user to make a choice. The player making the choice can pick any number remaining in the array — call it N. He then received N points. The opponent scores by claiming all factors of N remaining in the array for which he receives the sum of all of these factors.

Picking the largest prime first is usually a good choice since the opponent can then claim no points. In the case of the sample shown in Sample Run 1,

Sharpen your
factoring skills by
pitting yourself
against your
computer.

Herb chose 13; the computer could claim nothing since 13 is prime.

After each round of choosing and claiming, the computer reports the score and prints the array with all chosen and claimed numbers omitted.

When it's the user's turn to claim factors, the factors must be entered one at a time with a zero entered when no more factors are known. If the user misses any possible claims, the computer reports them after the user completes his turn.

Also, if the user tries to claim an illegal number, the computer will report why the move is illegal and add, "You lose your turn."

When all numbers have been removed from the array, the computer reports the final score and asks the user if he would like the opportunity to play again. If the user decides not to play anymore, the computer reports the number of games won by each player and stops.

Sample Run 2 shows a listing of another game and illustrates the computer's reaction to erroneous input.

The computer determines its choices in a simple but thorough way. It sums the remaining claimable factors of each of the numbers still in the array. The computer's move is that choice which

maximizes the difference between the number and the sum of its claimable factors. A dimensioned variable, A, stores the sum of the claimable factors of each number. Thus the computer's choice is determined by finding the largest remaining $I - A(I)$.

Program Listing 1 shows Factor as implemented on the DEC system 10 computer system. (Program Notes refer to this listing.)

Program Listing 2 shows Factor as written for Radio Shack TRS-80 Level I BASIC. It's actually a more convenient program to run, since the array remains on the screen and does not need to be rewritten each time. This feature uses the PRINT AT command. The program as listed makes use of abbreviations in Level I BASIC and omits remarks to conserve storage.

One modification in the TRS-80 version is that A(I) is set negative if the integer I has been removed from the array. This feature is necessary since the subscripted string used to test an integer's presence in the DEC system 10 version is not available on the TRS-80.

The TRS-80 program runs in 4K of RAM.

Ambitious factorers can easily expand this program to work beyond the limit of 81. Also, if you have a slow terminal, you will probably want to modify the DEC system 10 version so that it doesn't print the array each time. This modification puts the heavy burden of bookkeeping on you, but you may find it preferable to long waits for output.

If you find the computer is too good for you, you can tone down its game by replacing the choice of strategy by some random form of selection. If, on the other hand, you want a stiffer challenge from the computer, its game could be improved by implementing a look-ahead strategy for choice selection.

Happy factoring!

□

Program Listing 1

Decsystem 10 version of FACTOR

```

00010 RFM*** FACTOR BY HERR DERSEHEM.
00020 RFM*** INSTRUCTIONS FOR PLAYING ARE IN THE FOLLOWING
00030 RFM*** PRINT STATEMENTS.
00040 PRINT "THIS IS THE GAME OF FACTOR. YOU PLAY AGAINST"
00050 PRINT "THE COMPUTER. WHEN IT IS YOUR TURN, YOU CHOOSE"
00060 PRINT "A NUMBER FROM THE ARRAY OF NUMBERS DISPLAYED."
00070 PRINT "AFTER YOU HAVE CHOSEN A NUMBER, I CAN THEN CLAIM"
00080 PRINT "ALL NUMBERS REMAINING IN THE ARRAY WHICH ARE"
00090 PRINT "FACTORS OF YOUR CHOICE. FOR EXAMPLE, IF YOU"
00100 PRINT "CHOOSE 18, I CAN CLAIM 2,3,6, AND 9 IF THEY"
00110 PRINT "ARE STILL IN THE ARRAY. YOU WOULD RECEIVE 18"
00120 PRINT "POINTS AND I WOULD RECEIVE 20. THEN IT WOULD"
00130 PRINT "BE MY TURN TO CHOOSE AFTER WHICH YOU CLAIM FACTORS."
00140 PRINT "LET'S PLAY FACTOR!"
00150 DIM A(81),N$(81)
00160 RFM*** VARIABLES USED THROUGHOUT THE PROGRAM:
00170 RFM***      A$      PLAYER'S NAME
00180 RFM***      A(I)    SUM OF FACTORS OF I REMAINING
00190 RFM***      N$(I)   TWO CHARACTER STRING ELEMENT FOR
00200 RFM***              PRINTING. IT IS I IF I IS IN THE
00210 RFM***              ARRAY AND BLANK OTHERWISE.
00220 RFM***      A      PLAYER'S SCORE
00230 RFM***      B      COMPUTER'S SCORE
00240 RFM***      W      NUMBER OF COMPUTER WINS
00250 RFM***      Z      NUMBER OF PLAYER WINS
00260 PRINT
00270 W=0
00280 Z=0
00290 PRINT "WHAT IS YOUR NAME?";

```

(Continued on following page)

Program Notes

Statement numbers	Purpose
10-260	Introductory remarks and instructions for playing.
270-600	Initialization of variables and arrays.
650-1170	Computer's move.
650-740	Find the best choice.
750-800	Record choice.
810-950	Accept and validate any claims.
960-1010	Record claims.
1020-1100	Test for and notify of unclaimed factors.
1110-1150	Test for end of game.
1160-1170	Print board.
1180-1570	Player's move.
1180-1280	Accept and validate choice.
1290-1350	Record choice.
1360-1510	Find, print and record all claims.
1520-1530	Print board.
1540-1570	Test for end of game.
1580-1760	Print the final score and offer another game.
2000-2090	Subroutine to print the board.

Sample Run II

(User responses are underlined>)

WHAT IS YOUR NAME ?HERB
 HOW LARGE DO YOU WANT THE ARRAY (MAX 81) ?15
 DO YOU WANT TO CHOOSE FIRST (1=YES,0=NO) ?1

HERB'S SCORE 0 COMPUTER'S SCORE 0
 2 3 4 5 6 7 8 9 10
 11 12 13 14 15

YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?13
 I CLAIM NOTHING.

HERB'S SCORE 13 COMPUTER'S SCORE 0
 2 3 4 5 6 7 8 9 10
 11 12 14 15

MY TURN. I CHOOSE 11
 WHAT FACTOR OF 11 DO YOU CLAIM (0=NONE) ?0

HERB'S SCORE 13 COMPUTER'S SCORE 11
 2 3 4 5 6 7 8 9 10
 12 14 15

YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?15
 I CLAIM 3 5

HERB'S SCORE 28 COMPUTER'S SCORE 19
 2 4 6 7 8 9 10
 12 14

MY TURN. I CHOOSE 9
 WHAT FACTOR OF 9 DO YOU CLAIM (0=NONE) ?0

HERB'S SCORE 28 COMPUTER'S SCORE 28
 2 4 6 7 8 10
 12 14

YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?14
 I CLAIM 2 7

(Continued on following page)


```

00300 INPUT A$
00310 PRINT "HOW LARGE DO YOU WANT THE ARRAY (MAX 81)";
00320 INPUT N
00330 IF N>81 THEN 360
00340 IF N<2 THEN 360
00350 IF N=INT(N) THEN 380
00360 PRINT "ILLEGAL VALUE ENTERED. PLEASE TRY AGAIN."
00370 GOTO 310
00380 PRINT "DO YOU WANT TO CHOOSE FIRST (1=YES,0=NO)";
00390 INPUT C
00400 IF (C-1)*C = 0 THEN 440
00410 PRINT "PLEASE ENTER 1 OR 0."
00420 GOTO 380
00430 REM*** INITIALIZE A,B,A(),N$( )
00440 A=0
00450 B=0
00460 PRINT
00470 MAT A=ZER
00480 FOR I=2 TO N
00490   N$(I)=STR$(I)
00500   IF I>9 THEN 520
00510   N$(I)=" "+N$(I)
00520 NEXT I
00530 REM*** SUBROUTINE 2000 PRINTS THE ARRAY.
00540 GOSUB 2000
00550 REM*** SET EACH A(J)= SUM OF FACTORS OF J
00560 FOR I=2 TO N/2
00570   FOR J=2*I TO N STEP I
00580     A(J)=A(J)+I
00590   NEXT J
00600 NEXT I
00610 REM*** IF PLAYER MOVER FIRST, WE BRANCH HERE.
00620 IF C=1 THEN 1190
00630 REM*** COMPUTER'S MOVE - FIND THE REMAINING VALUE L
00640 REM*** WHICH GIVES THE GREATEST PROFIT AFTER CLAIMS.
00650 M=1000000
00660 L=0
00670 FOR I=2 TO N
00680   IF A(I)-I>M THEN 720
00690   IF N$(I)=" " THEN 720
00700   M=A(I)-I
00710   L=I
00720 NEXT I
00730 REM*** IF L=0 THEN NO CHOICES LEFT.
00740 IF L=0 THEN 1590
00750 PRINT "MY TURN. I CHOOSE ";L
00760 B=B+L
00770 FOR I=2*L TO N STEP L
00780   A(I)=A(I)-L
00790 NEXT I
00800 N$(L)=" "
00810 PRINT "WHAT FACTOR OF ";L;" DO YOU CLAIM (0=NONE)";
00820 REM*** ACCEPT CLAIM AND CHECK FOR VALIDITY.
00830 INPUT D
00840 IF D=0 THEN 1030
00850 IF D<>INT(D) THEN 870
00860 IF (D-1)*(D-N-1)<0 THEN 890
00870 PRINT "CHOICE OUT OF RANGE. YOU LOSE YOUR TURN."
00880 GOTO 1030
00890 IF N$(D)<>" " THEN 920
00900 PRINT D;" IS NOT AVAILABLE. YOU LOSE YOUR TURN."
00910 GOTO 1030
00920 IF INT(L/D)*D=L THEN 960
00930 PRINT D;" IS NOT A FACTOR OF ";L;". YOU LOSE YOUR TURN."

```

```

HERB'S SCORE 42      COMPUTER'S SCORE 37
   4      6      8      10
12
MY TURN. I CHOOSE 10
WHAT FACTOR OF 10 DO YOU CLAIM (0=NONE) ?0

HERB'S SCORE 42      COMPUTER'S SCORE 47
   4      6      8
12
YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?4
I CLAIM NOTHING.

HERB'S SCORE 46      COMPUTER'S SCORE 47
   6      8
12
MY TURN. I CHOOSE 8
WHAT FACTOR OF 8 DO YOU CLAIM (0=NONE) ?0

HERB'S SCORE 46      COMPUTER'S SCORE 55
   6
12

```

```

YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?12
I CLAIM 6
HERB'S SCORE 58      COMPUTER'S SCORE 61

FINAL SCORE - HERB 58      COMPUTER 61
I WON THAT TIME, BUT YOU CAN HAVE ANOTHER CHANCE.
WANT TO PLAY AGAIN (1=YES,0=NO) ?0
I WON 1 GAMES AND YOU WON 0

```

Sample Run 2

(User responses are underlined>)

```

WHAT IS YOUR NAME ?HERB
HOW LARGE DO YOU WANT THE ARRAY ?35
DO YOU WANT TO CHOOSE FIRST (1=YES,0=NO) ?0
HERB'S SCORE 0      COMPUTER'S SCORE 0
  2  3  4  5  6  7  8  9  10

```



```

C0940 GOTO 1030
C0950 REM*** UPDATE A(I) FOR CHOICE D.
C0960 FOR I=2*D TO N STEP D
C0970   A(I)=A(I)-D
C0980 NEXT I
C0990 N$(D)=" "
C1000 A=A+D
C1010 GOTO 810
C1020 REM*** TEST IF ALL FACTORS CLAIMED.
C1030 IF A(L)<=0 THEN 1100
C1040 PRINT "YOU COULD ALSO HAVE CHOSEN ";
C1050 FOR I=2 TO L/2
C1060   IF N$(I)=" " THEN 1090
C1070   IF INT(L/I)*I<>L THEN 1090
C1080   PRINT I;" ";
C1090 NEXT I
C1100 N$(L)=" "
C1110 REM*** TEST FOR GAME OVER
C1120 FOR I=2 TO N
C1130   IF N$(I)<>" " THEN 1170
C1140 NEXT I
C1150 GOTO 1590
C1160 REM*** PRINT ARRAY.
C1170 GOSUB 2000
C1180 REM*** NOW IT'S PLAYER'S TURN.
C1190 PRINT
C1200 PRINT "YOUR TURN. WHAT NUMBER DO YOU CHOOSE";
C1210 INPUT C
C1220 REM*** TEST FOR VALID CHOICE.
C1230 IF (C-1)*(C-N-1)<0 THEN 1260
C1240 PRINT "CHOICE OUT OF RANGE. CHOOSE AGAIN."
C1250 GOTO 1200
C1260 IF N$(C)<>" " THEN 1300
C1270 PRINT C;" IS NOT AVAILABLE. CHOOSE AGAIN."
C1280 GOTO 1200
C1290 REM*** UPDATE A(I) FOR THIS CHOICE C.
C1300 FOR I=2*C TO N STEP C
C1310   A(I)=A(I)-C
C1320 NEXT I
C1330 A=A+C
C1340 N$(C)=" "
C1350 L=C
C1360 PRINT "I CLAIM ";
C1370 REM*** ARE THERE ANY CLAIMS?
C1380 IF A(C)>0 THEN 1420
C1390 PRINT " NOTHING."
C1400 GOTO 1530
C1410 REM*** FIND AND PRINT ALL CLAIMS.
C1420 FOR I=2 TO L/2
C1430   IF N$(I)=" " THEN 1510
C1440   IF INT(L/I)*I<>L THEN 1510
C1450   PRINT I;
C1460   FOR J=2*I TO N STEP I
C1470     A(J)=A(J)-I
C1480   NEXT J
C1490   B=B+I
C1500   N$(I)=" "
C1510 NEXT I
C1520 REM*** PRINT THE ARRAY.
C1530 GOSUB 2000
C1540 REM*** TEST FOR GAME OVER.
C1550 FOR I=2 TO N
C1560   IF N$(I)<>" " THEN 650
C1570 NEXT I

```

(Continued on following page)

```

11 12 13 14 15 16 17 18 19
20 21 22 23 24 25 26 27
MY TURN. I CHOOSE 23
WHAT FACTOR OF 23 DO YOU CLAIM (0=NONE) ?11
11 IS NOT A FACTOR OF 23. YOU LOSE YOUR TURN.

HERP'S SCORE 0          COMPUTER'S SCORE 23
 2  3  4  5  6  7  8  9 10
11 12 13 14 15 16 17 18 19
20 21 22          24 25 26 27

YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?25
I CLAIM 5
HERP'S SCORE 25        COMPUTER'S SCORE 24
 2  3  4  5  6  7  8  9 10
11 12 13 14 15 16 17 18 19
20 21 22          24 26 27

MY TURN. I CHOOSE 19
WHAT FACTOR OF 19 DO YOU CHOOSE (0=NONE) ?0

HERP'S SCORE 25        COMPUTER'S SCORE 47
 2  3  4  5  6  7  8  9 10

```

```

11 12 13 14 15 16 17 18
20 21 22          24 26 27

YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?23
23 IS NOT AVAILABLE. CHOOSE AGAIN.
YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?27
I CLAIM 3 9
HERP'S SCORE 52        COMPUTER'S SCORE 59
 2  3  4  5  6  7  8  9 10
11 12 13 14 15 16 17 18
20 21 22          24 26

MY TURN. I CHOOSE 17
WHAT FACTOR OF 17 DO YOU CLAIM (0=NONE) ?0

HERP'S SCORE 52        COMPUTER'S SCORE 76
 2  3  4  5  6  7  8  9 10
11 12 13 14 15 16 17 18
20 21 22          24 26

YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?18
I CLAIM 2 6

```

(Continued on following page)


```

01580 REM*** GAME'S OVER - PRINT RESULT.
01590 PRINT "FINAL SCORE - ",A$;" ";A,"COMPUTER ";B
01600 IF A>B THEN 1720
01610 IF A=B THEN 1750
01620 W=W+1
01630 PRINT "I WON THAT TIME, BUT YOU CAN HAVE ANOTHER CHANCE."
01640 PRINT "WANT TO PLAY AGAIN (1=YES,0=NO)";
01650 INPUT C
01660 IF C=1 THEN 310
01670 IF C=0 THEN 1700
01680 PRINT "PLEASE ENTER 1 OR 0."
01690 GOTO 1640
01700 PRINT "I WON ";W;" GAMES AND YOU WON ";Z
01710 STOP
01720 Z=Z+1
01730 PRINT "YOU WON, BUT YOU WERE LUCKY."
01740 GOTO 1640
01750 PRINT "TIE GAME."
01760 GOTO 1640
01990 REM*** PRINT ARRAY
02000 PRINT
02010 PRINT A$;"S SCORE ";A;" ";B;"COMPUTER'S SCORE ";B
02020 FOR I=2 TO N
02030 PRINT N$(I);" ";
02040 IF INT((I-1)/9)*9<>I-1 THEN 2060
02050 PRINT
02060 NEXT I
02070 PRINT
02080 RETURN
02090 END

```

Program Listing 2

TRS-80 version of factor

```

40 P. "THIS IS THE GAME OF FACTOR. YOU PLAY AGAINST"
50 P. "THE COMPUTER. WHEN IT IS YOUR TURN, YOU CHOOSE"
60 P. "A NUMBER FROM THE ARRAY OF NUMBERS DISPLAYED."
70 P. "AFTER YOU HAVE CHOSEN A NUMBER, I CAN THEN CLAIM"
80 P. "ALL NUMBERS REMAINING IN THE ARRAY WHICH ARE"
90 P. "FACTORS OF YOUR CHOICE. FOR EXAMPLE, IF YOU"
100 P. "CHOOSE 18, I CAN CLAIM 2,3,6, AND 9 IF THEY"
110 P. "ARE STILL IN THE ARRAY. YOU WOULD RECEIVE 18"
120 P. "POINTS AND I WOULD RECEIVE 20. THEN IT WOULD"
130 P. "BE MY TURN TO CHOOSE AFTER WHICH YOU CLAIM FACTORS."
140 P. "LET'S PLAY FACTOR!"
260 P. :W=0:Z=0
290 P. "WHAT IS YOUR NAME";:IN. A$
310 P. "HOW LARGE DO YOU WANT THE ARRAY (MAX. 81)";:IN. N
330 IF (N>1)*(N<82)*(N=INT(N)) THEN 380
340 P. "ILLEGAL VALUE ENTERED. PLEASE TRY AGAIN.";G. 310
380 P. "DO YOU WANT TO CHOOSE FIRST (1=YES,0=NO)";:IN. C
400 IF (C=1)+(C=0) THEN 440
410 P. "PLEASE ENTER 1 OR 0";G. 380
440 A=0:B=0:CLS
450 P. AT 0,A$;"S SCORE ";A;
460 P. AT 32,"TRS-80'S SCORE ";B
470 F. I=2 TO N:A(I)=0:P. AT 48+8*I,I;N. I
560 P. I=2 TO N/2:F J=2*I TC N STEP I:A(J)=A(J)+I:N. J:N. I
620 IF C=1 THEN 1170
650 M=1000000:L=0:F. I=2 TO N
680 IF (A(I)-I>M)+(A(I)<0) THEN 720
700 M=A(I)-I:L=I
720 N. I

```

```

HERB'S SCORE 70      COMPUTER'S SCORE 84
  4      7  8      10
11 12 13 14 15 16
20 21 22      24 26
MY TURN. I CHOOSE 15
WHAT FACTOR OF 15 DO YOU CLAIM (0=NONE) ?0

```

```

HERB'S SCORE 70      COMPUTER'S SCORE 99
  4      7  8      10
11 12 13 14      16
20 21 22      24 26

```

```

YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?26
I CLAIM 13
HERB'S SCORE 96      COMPUTER'S SCORE 112
  4      7  8      10
11 12      14      16
20 21 22      24
MY TURN. I CHOOSE 21
WHAT FACTOR OF 21 DO YOU CLAIM (0=NONE) ?0

```

```

YOU COULD ALSO HAVE CHOSEN 7
HERB'S SCORE 96      COMPUTER'S SCORE 133
  4      7  8      10
11 12      14      16
20      22      24

```

```

YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?22
I CLAIM 11
HERB'S SCORE 118      COMPUTER'S SCORE 144
  4      7  8      10
      12      14      16
20      24

```

```

MY TURN. I CHOOSE 10
WHAT FACTOR OF 10 DO YOU CLAIM (0=NONE) ?0
HERB'S SCORE 118      COMPUTER'S SCORE 154
  4      7  8
      12      14      16
20      24

```



```

740 IF L=0 THEN 1590
750 P. AT 704,"MY TURN. I CHOOSE ";L
760 B=B+L:P. AT 32,"TRS-80'S SCORE ";B
770 IF 2*L>N THEN 800
780 P. I=2*L TO N STEP L:A(I)=A(I)-L:N. I
800 P. AT 48+8*L," ";
810 P. AT 832:P. AT 832,"WHAT FACTOR OF";L;"DO YOU CLAIM (0=NONE)";
830 IN. D:IF D=0 THEN 1030
850 IF (D>1)*(D<=N)*(D=INT(D)) THEN 890
870 P. AT 896,"CHOICE OUT OF RANGE.";
880 P. " YOU LOSE YOUR TURN.":F. I=1 TO 1500:N. I:G. 1030
890 IF A(D)>0 THEN 920
900 P. AT 896,D;"IS NOT AVAILABLE.":G. 880
920 IF INT(L/D)*D=L THEN 960
940 P. AT 896,D;"IS NOT A FACTOR OF";L;"":G. 880
960 IF 2*D>N THEN 990
970 F. I=2*D TO N STEP D:A(I)=A(I)-D:N. I
990 A(D)=-100:P. AT 48+8*D," ";
1000 A=A+D:P. AT 0,A$;"S SCORE ";A;
1010 G. 810
1030 IF A(L)=0 THEN 1100
1040 P. AT 896:P. AT 896,"YOU COULD ALSO HAVE CHOSEN ";
1050 F. I=2 TO L/2:IF (A(I)<0)+(INT(L/I)*I<>L) THEN 1090
1080 P. I;" ";
1090 N. I:F. I=1 TO 1500:N. I
1100 A(L)=-100
1120 F. I=2 TO N:IF A(I)>0 THEN 1170
1140 N. I:G. 1590
1170 P. AT 768:P.:P.
1200 P. AT 704:P. AT 704,"YOUR TURN. WHAT NUMBER DO YOU CHOOSE";
1210 IN. C
1230 IF (C>1)*(C<=N)*(C=INT(C)) THEN 1260
1240 P. AT 769,"CHOICE OUT OF RANGE. CHOOSE AGAIN.":G. 1200
1260 IF A(C)>0 THEN 1300
1270 P. AT 768,C;"IS NOT AVAILABLE. CHOOSE AGAIN.":G. 1200
1300 IF 2*C>N THEN 1320
1310 F. I=2*C TO N STEP C:A(I)=A(I)-C:N. I
1320 P. AT 768
1330 A=A+C:P. AT 0,A$;"S SCORE ";A;
1340 P. AT 48+8*C," ";
1350 L=C:P. AT 832,"I CLAIM ";
1370 O=840
1380 IF A(C)>0 THEN 1420
1390 P. "NOTHING.":G. 1530
1420 F. I=2 TO L/2:IF (A(I)<0)+(INT(L/I)*I<>L) THEN 1510
1450 P. AT 0,I;" ":O=O+4:F. J=I TO N STEP I:A(J)=A(J)-I:N. J
1490 B=B+I:P. AT 32,"TRS-80'S SCORE ";B:A(I)=-100
1500 P. AT 48+8*I," ";
1510 N. I
1530 A(C)=-100
1540 F. I=2 TO N:IF A(I)>0 THEN 650
1570 N. I
1590 CLS:P. "FINAL SCORE - ";A$;" ";A,"TRS-80 ";B
1600 IF A>B THEN 1720
1610 IF A=B THEN 1750
1620 W=W+1:P. "I WON THAT TIME, BUT YOU CAN HAVE ANOTHER CHANCE."
1630 P. "WANT TO PLAY AGAIN (1=YES,0=NO)";:IN. C
1660 IF C=1 THEN 310
1670 IF C=0 THEN 1700
1680 P. "PLEASE ENTER 1 OR 0.":G. 1630
1700 P. "I WON ";W;" GAMES AND YOU WON ";Z
1710 STOP
1720 Z=Z+1
1730 P. "YOU WON, BUT YOU WERE LUCKY.":G. 1630
1750 P. "TIP GAME.":G. 1630

```

YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?20

I CLAIM 4
HERE'S SCORE 138

12 14 16 7 8 24

MY TURN. I CHOOSE 12
WHAT FACTOR OF 12 DO YOU CLAIM (0=NONE) ?6
6 IS NOT AVAILABLE. YOU LOSE YOUR TURN.

HERE'S SCORE 138 COMPUTER'S SCORE 170

14 16 7 8 24

YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?24

I CLAIM 8
HERE'S SCORE 162 COMPUTER'S SCORE 178

14 16 7 8

MY TURN. I CHOOSE 16

WHAT FACTOR OF 16 DO YOU CLAIM (0=NONE) ?2

HERE'S SCORE 162 COMPUTER'S SCORE 194

14 7

YOUR TURN. WHAT NUMBER DO YOU CHOOSE ?14

I CLAIM 7
HERE'S SCORE 176 COMPUTER'S SCORE 201

FINAL SCORE - HERB 176 COMPUTER 201
I WON THAT TIME, BUT YOU CAN HAVE ANOTHER CHANCE.
WANT TO PLAY AGAIN (1=YES,0=NO) ?0
I WON 1 GAMES AND YOU WON 0